INTRODUCTION

The AXIS and LEGEND statements allow users to control many aspects of their graphs in Version 5 of SAS/GRAPH software. The examples in this paper illustrate some of the main features. Users should be able to adapt some of the following examples to meet their own needs.

DEFINITIONS

In this paper parts of legends and axes are described with the following terminology. See Figure 1 for some examples.

Value
The text associated with a value of the variable represented on the axis or legend.
Label
The text associated with the name of the variable represented on the axis or legend.
Major tick mark
A mark along an axis associated with a value.
Minor tick mark
A mark between major tick marks.
Offset
Distance between the first or last major tick mark and the end of the axis line.
Axis line
The line running the length of the axis. The tick marks are perpendicular to the axis line.
Axis origin
The coordinates or the left or lower end of the axis line.
Legend shape
The bars, lines, or symbols in a legend.

SPECIFYING AND REFERENCING

Using the AXIS and LEGEND statements involves specifying then referring to the description created with the statement. Specifying an AXIS or LEGEND statement does not cause a graph to generate an axis or legend; instead it allows users to provide a precise description for their graph. Example:

```
PROC GPLOT;
AXIS1 LABEL=NONE;
PLOT Y * X / HAXIS=AXIS1;
```

CONTROLLING FONTS AND HEIGHTS

The font used for the text on an axis or legend can be specified with the F=fontname option of the VALUE= and LABEL= options. Example 1 shows how to specify the XSWISS font to be used for both the labels and the values on both axes and in the legend. Notice the H= option that increases the size of the text to help make it more readable.

CONTROLLING COLORS

There are several ways to specify the colors of text and other parts of an axis and legend. The general rule of which color takes precedence is that a color referring to a more specific part of the graph overrides a more general color. For instance, on an AXIS statement the COLOR= option is the most general. The CTEXT= or CAXIS= options from the PLOT (etc.) statement are more specific and therefore override COLOR= when they are specified. An even more specific color specification is to use the C= option in the VALUE= or LABEL= options of the AXIS statement. So specifying:

```
PROC GPLOT;
AXIS1 COLOR=BLACK
LABEL=(C=RED);
PLOT Y*X / CTEXT=BLUE HAXIS=AXIS1 VAXIS=AXIS1;
```

would cause all of the axis to be BLACK (COLOR=BLACK) except for the text which would be BLUE (CTEXT=BLUE) except for the labels which would be RED (VALUE=(C=RED)).

USING QUOTED STRINGS

Quoted strings can be used in the VALUE= and LABEL= options of the AXIS and LEGEND statements to specify the exact text to be used. The number of quoted strings does not determine the number of values. If no T= options (see CONTROLLING INDIVIDUAL VALUES) are specified the quoted strings are matched one-to-one with the values. Example 2 shows how to use quoted strings without T=.

CONTROLLING INDIVIDUAL VALUES

The T= option of the VALUE= option of AXIS and LEGEND statements allows users to specify different options for one or more of the values on an axis or legend. For example to make the fifth value on an axis to be red while the other values are blue specify:

```
AXIS1 VALUE=(C=BLUE T=5 C=RED);
```

Note that since C=BLUE appears before any T= option it refers to all values except when overridden after a T= option.

CONTROLLING MAJOR AND MINOR TICK MARKS

The MAJOR= and MINOR= options of the AXIS statement allow users to describe the appearance of the major and minor tick marks, respectively,
along an axis. The marks can be suppressed by specifying MAJOR=NONE MINOR=NONE or users can control the height (or length) of the marks (H=), the color (C=) and the number (N=). It should be noted that specifying:

\[ \text{AXIS MAJOR=(N=5) ORDER=0 to 10;} \]

will produce an axis with 11 major tick marks since the ORDER= option overrides the N= in the MAJOR= option. In example 4, the N= option of the MAJOR= and MINOR= options is used on the vertical axes to control the number of tick marks.

CONTROLLING AXIS ORIGIN

The origin of an axis can be partially or completely specified with the ORIGIN= option of the AXIS statement. The coordinates of the origin can be specified in any combination of inches, centimeters, percent, or cells. For example:

\[ \text{AXIS ORIGIN=(10 cm, 10 cm);} \]
\[ \text{AXIS ORIGIN=(20 pct);} \]
\[ \text{AXIS ORIGIN=(10 cells);} \]

Example 4 shows the use of ORIGIN=.

CONTROLLING AXIS LENGTH

The length of an axis can be specified in inches, centimeters, percent, or cells. This can be useful to create a series of graphs that will be the same size or to produce a graph of a fixed size to be placed in a document. Example 4 shows the use of LENGTH=.

CONTROLLING AXIS OFFSETS

An axis offset is the distance between the first (or last) major tick mark and the end of the axis line. In PROC GPLOT the offsets are normally 1/2 of a character cell to allow room for symbols in the graph. The most useful specification of offsets is:

\[ \text{AXIS OFFSET=(0);} \]

This will give a graph a "cleaner" look when there are no symbols. If the OFFSET= option is used on an axis statement later used to describe a midpoint axis (MAXIS=) from a HBAR or VBAR statement in PROC GCHART, care must be taken so that the offset specified leaves enough room for the first and last bars to fit along the axis. Example 4 shows the use of OFFSET=.

CONCLUSION

This paper contains some helpful hints for using AXIS and LEGEND statements and explains how many of the options work and interact. For more details consult the SAS/GRAPH® User’s Guide Version 5 Edition.

SAS® and SAS/GRAPH® are registered trademarks of SAS Institute Inc., Cary, NC, USA.

---

CODE FOR EXAMPLE 1

```sas
title f=XSWISSB 'Specifying Fonts';
footnote f=xswiss 'Example 1';
proc gchart data=AUTO;
   legend1 label=( f=XSWISS) value=(f=XSWISS);
   axis1 label=(f=XSWISS) value=( f=XSWISS);
   hbar MODEL / subgroup = MAKE
      sumvar = SALES
      legend = LEGEND1
      raxis = AXIS1
      maxis = AXI1S
      nostats frame;
run;
```

CODE FOR EXAMPLE 2

```sas
title f=XSWISSB 'Using Quoted Strings';
footnote f=xswiss 'Example 2';
proc gchart data=ENERGY;
   vbar SOURCE / descending
      sumvar=PROD
      maxis=AXIS2
      raxis=AXIS1
      frame;
   axis1 order=0 to 20 by 5
      label=(f=xswiss 'Quadrillion'
            J=R 'Btu')
      value=(f=xswiss) ;
   axis2 value=(f=xswiss 'Coal' 'Oil' 'Gas')
      label=NONE;
run;
```

CODE FOR EXAMPLE 3

```sas
title f=XSWISSB 'Using T=';
footnote f=xswiss 'Example 3';
proc gchart data=COMPUTER;
   vbar company / sumvar=sales
      maxis=AXIS2
      raxis=AXIS1
      frame;
   axis1 label=(f=XSWISS '$100K')
      value=( f=XSWISS)
      order=O to 750 by 250;
   axis2 value=(f=XSWISS
      t='1 Fujitsu'
      j=C f=kan45 '84'X
      t='2 Hitachi'
      j=C f=kan45 '62D'X
      t='3 IBM')
      label=NONE;
run;
```
CODE FOR EXAMPLE 4

goption nodisplay;
title; footnote;
symbol i=JOIN v=NONE;
proc gplot data=EXCHANGE gout=EXCH;
   format DATE worddate3.;
   axis1 length=20pct
      origin=(20pct, 10pct)
      major=(N=5)
      value=(f=xswiss)
      label=(a=90 f=xswiss)
      minor=NONE;
   axis2 offset=(0)
      length=70pct
      origin=(20pct, 10pct)
      order='01OCT85'd to '01FEB86'd by month
      value=(f=xswiss);
   plot POUND * DATE / vaxis=AXIS1
      haxis=AXIS2
      name='POUND'
      frame;
   axis3 length=20pct
      origin=(20pct, 30pct)
      major=(N=5)
      value=(f=xswiss)
      label=(a=90 f=xswiss)
      minor=NONE;
   axis4 offset=(0)
      length=70pct
      origin=(20pct, 30pct)
      order='01OCT85'd to '01FEB86'd by month
      label=NONE
      value=NONE;
   plot FRANC * DATE / vaxis=AXIS3
      haxis=AXIS4
      name='FRANC'
      frame;
   axis5 length=20pct
      origin=(20pct, 50pct)
      major=(N=5)
      value=(f=xswiss)
      label=(a=90 f=xswiss)
      minor=NONE;
   axis6 offset=(0)
      length=70pct
      origin=(20pct, 50pct)
      order='01OCT85'd to '01FEB86'd by month
      label=NONE
      value=NONE;
   plot MARK * DATE / vaxis=AXIS5
      haxis=AXIS6
      name='MARK'
      frame;
   axis7 length=20pct
      origin=(20pct, 70pct)
      major=(N=5)
      value=(f=xswiss)
      label=(a=90 f=xswiss)
      minor=NONE;
   axis8 offset=(0)
      length=70pct
      origin=(20pct, 70pct)
      order='01OCT85'd to '01FEB86'd by month
      label=NONE
      value=NONE;
   plot YEN * DATE / vaxis=AXIS7
      haxis=AXIS8
      name='YEN'
      frame;
run;
title f=xswiss 'Complex Axes';
footnote f=xswiss 'Example 4';
proc gslide gout=EXCH name='TITLE'; run;
goption display;

proc greplay igout=EXCH nofs;
tdef ALL 1 / llx=0 ulx=100 lrx=100
      lly=0 uly=100 urx=100 lry=0;
template ALL;
tplay 1:TITLE 1:POUND 1:FRANC 1:MARK 1:YEN;
end;

**AXIS Terms**

**Specifying Fonts**

**Using Quoted Strings**

---

Example 1

Example 2
Example 3

Complex Axes